Q:

Populate dates in a label based on dates in a hidden sheet using IIF

The code below uses information saved in a hidden worksheet based on values from another sheet with a codename of GeneralInfo and is called when specific cells in GeneralInfo are changed.  I think there would be a more efficient way to write this code.  the Processing.Label refers to labels on a worksheet codename of Processing.
Option Explicit
Sub PopulateServiceDates()
    Dim InHouseTitle As String, InHouseEval As String, ApprRev As String, ClsSentRev As String, HMDALast As String
    Dim BSASaved As String, oAppraisal As String, oTitle As String, IHTitleRec As String, IHEvalRec As String, ApprRec As String
    Dim TitleRec As String

    On Error GoTo ErrHandler

    InHouseTitle = SheetData.Range("Date_In_House_Title_Work_Ordered")
    InHouseEval = SheetData.Range("Date_In_House_Eval_Sent")
    ApprRev = SheetData.Range("Date_Appraisal_Review_Sent")
    ClsSentRev = SheetData.Range("Date_Docs_Sent_for_Review")
    HMDALast = SheetData.Range("HMDAClicked")
    BSASaved = SheetData.Range("Date_BSA_Saved")
    oAppraisal = SheetData.Range("Date_Outside_Appraisal_Ordered")
    oTitle = SheetData.Range("Date_Outside_Title_Ordered")
    IHTitleRec = SheetData.Range("Date_In_House_Title_Received")
    IHEvalRec = SheetData.Range("Date_In_House_Appraisal_Received")
    ApprRec = SheetData.Range("Date_Outside_Appraisal_Received")
    TitleRec = SheetData.Range("Date_Outside_Title_Received")

    Dim cap As String, cap2 As String, cap3 As String, cap4 As String, cap5 As String, cap6 As String
    cap = "In-House Title Work Ordered:" & vbCrLf & _
          "In-House Evaluation Ordered:" & vbCrLf & _
          "Appraisal Ordered:" & vbCrLf & _
          "Title Work Ordered:"

    cap2 = IIf(ApprRev <> vbNullString, ApprRev, "Not sent yet") & vbCrLf & _
           IIf(ClsSentRev <> vbNullString, ClsSentRev, "Not sent yet") & vbCrLf & _
           IIf(HMDALast <> vbNullString, HMDALast, "HMDA Helper not saved yet.") & vbCrLf & _
           IIf(BSASaved <> vbNullString, BSASaved, "N/A")

    cap3 = IIf(InHouseTitle <> vbNullString, InHouseTitle, "N/A") & vbCrLf & _
           IIf(InHouseEval <> vbNullString, InHouseEval, "N/A") & vbCrLf & _
           IIf(oAppraisal <> vbNullString, oAppraisal, "N/A") & vbCrLf & _
           IIf(oTitle <> vbNullString, oTitle, "N/A")

    cap4 = "Appraisal Sent for Review:" & vbCrLf & _
           "Closing Docs Sent for Review:" & vbCrLf & _
           "HMDA Helper Last Updated:" & vbCrLf & _
           "BSA Saved:"

    cap5 = "In-House Title Work Received:" & vbCrLf & _
          "In-House Evaluation Received:" & vbCrLf & _
          "Appraisal Received:" & vbCrLf & _
          "Title Work Received:"

    cap6 = IIf(IHTitleRec <> vbNullString, IHTitleRec, "N/A") & vbCrLf & _
           IIf(IHEvalRec <> vbNullString, IHEvalRec, "N/A") & vbCrLf & _
           IIf(ApprRec <> vbNullString, ApprRec, "N/A") & vbCrLf & _
           IIf(TitleRec <> vbNullString, TitleRec, "N/A")

    With Processing.Label12
        .caption = cap
        .Font.Size = 14
        .TextAlign = fmTextAlignRight
    End With

    With Processing.Label13
        .caption = cap3
        .Font.Size = 14
        .TextAlign = fmTextAlignLeft
    End With

    With Processing.Label9
        .caption = cap4
        .Font.Size = 14
        .TextAlign = fmTextAlignRight
    End With

    With Processing.Label11
        .caption = cap2
        .Font.Size = 14
        .TextAlign = fmTextAlignLeft
    End With

    With Processing.Label6
        .caption = cap5
        .Font.Size = 14
        .TextAlign = fmTextAlignRight
    End With

    With Processing.Label14
        .caption = cap6
        .Font.Size = 14
        .TextAlign = fmTextAlignLeft
    End With

ErrHandler:
    If Err.Number <> 0 Then
        If MsgBox("Error " & str(Err.Number) & " " & Err.Description & vbCrLf & _
                  "occured in the PopulateServiceDates Subroutine " & vbCrLf & _
                  "Please let Zack know what caused the error before clicking OK.", vbOKOnly + vbCritical, UCase("error")) = vbOK Then
            Resume Next
        End If
    End If

End Sub

A:

Declare your variables right before you use them. For your cap variables, assuming these stand for caption (more on that with descriptive variables), you have a Declare, Populate, Use in a 1-2-3 pattern.
Dim foo As String '<-- 1 Declare

'Code that does stuff

foo = "bar" '<-- 2 Populate

'More code

buzz.Caption = foo `<-- 3 Use

You can, and IMO should, reduce down to the assigning the property all at once without the variable.
buzz.Caption = "bar"

Using a default member hides what's going on, making your code harder to read. Avoid using them. The code below is using the default member on the Range object.
Dim defaultMemberAccess As String
defaultMemberAccess = SheetData.Range("Foo")

Be explicit about what you want it to obtain like as below. You now know it's accessing the Value2 member of the Range object.
Dim explicitlyStatedMemberAccess as String
explicitlyStatedMemberAccess = SheetData.Range("Foo").Value2

Explicitly qualify your Sub. Omitting any access modifier means its implicitly Public. Add that Public in to have Public Sub PopulateServiceDates() so it's obvious that it was intentional.

Variable names. Use descriptive variables. IHEvalRec made me immediately think it was in interface. That's because an I prefix by convention indicates interface. If I saw inHouseAppraisalReceivedText that would be a lot more informative and take less guesswork. Future-you will also thank you when you come back to this code.
This ties in to Label12, Label13, etc... Rename them labels because they aren't helping readability. Again future-you will thank you.

Magic numbers. 14 has what meaning? Is that the default font size you want to use? We already went over that and this ties in as well. Const DefaultFontSize As Double = 14 declares a constant and lets you know you want it to be the default font size so that .Font.size = DefaultFontSize is self describing.

Usage of IIF feels odd, like your sometimes expecting an error. You're accessing named ranges and it should be a safe assumption they are there. I'm of the opinion they should be behind a worksheet property, and that the property should handle what occurs if there's a barf error. Something like
Private Property Get InHouseTitleWorkOrderedText() As String
    Dim temp As String
    temp = SheetData.Range("Date_In_House_Title_Work_Ordered").Value2
    If temp = vbNullString Then
        InHouseTitleWorkOrderedText = "N/A"
    Else
        InHouseTitleWorkOrderedText = temp
    End If
End Property

so that when you are populating the labels caption property you end up with 
    With Processing.Label13
        .Caption = InHouseTitleWorkOrderedText & vbCrLf & _
                    InHouseEvaluationSentText & vbCrLf & _
                    OutsideAppraisalOrderedText & vbCrLf & _
                    OutsideTitleOrderedText
        .Font.Size = 14
        .TextAlign = fmTextAlign.fmTextAlignLeft
    End With

Once you have your properties set up like that you begin think about whether the property should doing any logic checking in the first place. Can that be done in the cell itself? If yes, the property reduces down to
Private Property Get InHouseTitleWorkOrderedText() As String
    InHouseTitleWorkOrderedText = SheetData.Range("Date_In_House_Title_Work_Ordered").Value2
End Property

Lastly and 2 minor things Rubberduck, to which I'm a contributor, helped me pick up that I forgot. You can use replace Str(Err.Number) with Str$(Err.Number) and UCase("error") with UCase$("error"). These are the alternate typed functions.
In truth however UCase$("error") should be just "ERROR".

Combining all the above I came up with
'SheetData
Public Property Get InHouseTitleWorkOrderedText() As String
    InHouseTitleWorkOrderedText = SheetData.Range("Date_In_House_Title_Work_Ordered").Value2
End Property

Public Property Get InHouseEvaluationSentText() As String
    InHouseEvaluationSentText = SheetData.Range("Date_In_House_Eval_Sent").Value2
End Property

Public Property Get AppriasalReviewSentText() As String
    AppriasalReviewSentText = SheetData.Range("Date_Appraisal_Review_Sent").Value2
End Property

Public Property Get DocumentSentForReviewText() As String
    DocumentSentForReviewText = SheetData.Range("Date_Docs_Sent_for_Review").Value2
End Property

Public Property Get HMDALastText() As String
    HMDALastText = SheetData.Range("HMDAClicked").Value2
End Property

Public Property Get BSASavedText() As String
    BSASavedText = SheetData.Range("Date_BSA_Saved")
End Property

Public Property Get OutsideAppraisalOrderedText() As String
    OutsideAppraisalOrderedText = SheetData.Range("Date_Outside_Appraisal_Ordered")
End Property

Public Property Get OutsideTitleOrderedText() As String
    OutsideTitleOrderedText = SheetData.Range("Date_Outside_Title_Ordered")
End Property

Public Property Get InHouseTitleReceivedText() As String
    InHouseTitleReceivedText = SheetData.Range("Date_In_House_Title_Received")
End Property

Public Property Get InHouseAppraisalReceivedText() As String
    InHouseAppraisalReceivedText = SheetData.Range("Date_In_House_Appraisal_Received")
End Property

Public Property Get OutsideAppraisalReceivedText() As String
    OutsideAppraisalReceivedText = SheetData.Range("Date_Outside_Appraisal_Received")
End Property

Public Property Get OutsideTitleReceivedText() As String
    OutsideTitleReceivedText = SheetData.Range("Date_Outside_Title_Received")
End Property

And the rewritten module
Module1
Option Explicit

Public Sub PopulateServiceDates()
    Const DefaultFontSize As Double = 14

    On Error GoTo ErrHandler

    With Processing.Label12
        .Caption = "In-House Title Work Ordered:" & vbCrLf & _
                    "In-House Evaluation Ordered:" & vbCrLf & _
                    "Appraisal Ordered:" & vbCrLf & _
                    "Title Work Ordered:"
        .Font.Size = DefaultFontSize
        .TextAlign = fmTextAlign.fmTextAlignRight
    End With

    With Processing.Label13
        .Caption = SheetData.InHouseTitleWorkOrderedText & vbCrLf & _
                    SheetData.InHouseEvaluationSentText & vbCrLf & _
                    SheetData.OutsideAppraisalOrderedText & vbCrLf & _
                    SheetData.OutsideTitleOrderedText
        .Font.Size = DefaultFontSize
        .TextAlign = fmTextAlign.fmTextAlignLeft
    End With

    With Processing.Label9
        .Caption = "Appraisal Sent for Review:" & vbCrLf & _
                    "Closing Docs Sent for Review:" & vbCrLf & _
                    "HMDA Helper Last Updated:" & vbCrLf & _
                    "BSA Saved:"
        .Font.Size = DefaultFontSize
        .TextAlign = fmTextAlign.fmTextAlignRight
    End With

    With Processing.Label11
        .Caption = SheetData.AppriasalReviewSentText & vbCrLf & _
                    SheetData.DocumentSentForReviewText & vbCrLf & _
                    SheetData.HMDALastText & vbCrLf & _
                    SheetData.BSASavedText
        .Font.Size = DefaultFontSize
        .TextAlign = fmTextAlign.fmTextAlignLeft
    End With

    With Processing.Label6
        .Caption = "In-House Title Work Received:" & vbCrLf & _
                    "In-House Evaluation Received:" & vbCrLf & _
                    "Appraisal Received:" & vbCrLf & _
                    "Title Work Received:"
        .Font.Size = DefaultFontSize
        .TextAlign = fmTextAlign.fmTextAlignRight
    End With

    With Processing.Label14
        .Caption = SheetData.InHouseTitleReceivedText & vbCrLf & _
                    SheetData.InHouseAppraisalReceivedText & vbCrLf & _
                    SheetData.OutsideAppraisalReceivedText & vbCrLf & _
                    SheetData.OutsideTitleReceivedText
        .Font.Size = DefaultFontSize
        .TextAlign = fmTextAlign.fmTextAlignLeft
    End With

ErrHandler:
    If Err.Number <> 0 Then
        If MsgBox("Error " & Str$(Err.Number) & " " & Err.Description & vbCrLf & _
                  "occured in the PopulateServiceDates Subroutine " & vbCrLf & _
                  "Please let Zack know what caused the error before clicking OK.", vbOKOnly + vbCritical, UCase$("error")) = vbOK Then
            Resume Next
        End If
    End If
End Sub
```

